



دانشکده مهندسی برق

جزوه :

تجزیه و تحلیل مدارهای الکتریکی و الکترونیکی

HSpice با

تهیه و تنظیم از :

سیروس طوفان

زمستان ۸۳

با گسترش و پیچیدگی روزافزون مدارهای الکترونیکی استفاده از کامپیوتر در تجزیه و تحلیل اینگونه مدارها از ضرورت برخوردار است. کامپیوتر کمک می‌کند تا بتوانیم بینش خود را در موارد زیر، درباره مدارهای الکترونیکی تعمیق دهیم :

- ۱- بررسی اثر تغییرات اجزاء مدار و محاسبه حساسیت رفتار مدار نسبت به آن (شامل مقاومت‌ها، سلفها، خازن‌ها، دیودها و ترانزیستورها و ... )
- ۲- درنظر گرفتن آثار غیر خطی المان‌ها در رفتار مدار
- ۳- بهینه‌کردن مدار با توجه به موارد فوق
- ۴- بررسی رفتار مدار از نظر نویز و اعوجاج‌ها و همچنین از نظر تجزیه و تحلیل در حوزه فرکانس و بدست آوردن طیف فرکانسی بر مبنای تجزیه و تحلیل فوریه بدون اینکه نیاز به استفاده از دستگاه‌های گران قیمتی مانند Noise Figure Meter و یا Spectrum Analyzer داشته باشیم .

### برنامه HSpice

برنامه‌ای برای شبیه‌سازی مدارهای الکتریکی و الکترونیکی می‌باشد. این برنامه قابلیت تجزیه و تحلیل‌های زیر را در مدارهای الکتریکی و الکترونیکی دارد .

- ۱- تجزیه و تحلیل dc
- ۲- تجزیه و تحلیل گذرا
- ۳- تجزیه و تحلیل نویز
- ۴- تجزیه و تحلیل ac
- ۵- تجزیه و تحلیل فوریه

در برنامه HSpice مدارها می‌توانند شامل مقاومت، خازن، سلف ( بدون تزویج متقابل و با تزویج متقابل ) کلیدهای کنترل شده با جریان یا ولتاژ، منابع ولتاژ و جریان مستقل و وابسته و عناصر نیمه‌هادی مانند دیود، ترانزیستور، JFET و MESFET باشند. برنامه HSpice قادر است خطوط انتقال را هم بعنوان یک عنصر مداری بررسی نماید.

در استفاده از برنامه HSpice برای تجزیه و تحلیل مدارها چهار مرحله اساسی زیر را باید در نظر گرفت:

- ۱- تبدیل مدار به یک برنامه (file Netlist) قابل فهم برای HSpice
- ۲- تعیین نوع تجزیه و تحلیل مورد نیاز

-۳- تعیین ورودی و خروجی‌های مورد نظر

-۴- اجرا کردن برنامه و دیدن شکل موجهای نقاط مختلف مدار


### تخصیص شماره یا اسم به گره

هر مدار شامل چند گره و چند شاخه و یک یا چند ورودی و یک یا چند خروجی خواهد بود. گرهها به دو صورت مشخص می‌شوند: با نام (حرف یا کلمه) یا با شماره (عدد). در نامگذاری گرهها بصورت غیر عددی، نام گرهها باید بیشتر از ۱۶ کاراکتر باشد و حتماً "با یکی از حروف الفبا آغاز گردد. در نامگذاری می‌توان از کاراکترهای زیر استفاده کرد.

! # \$ % \* + - / < > [ ] \_

گرهای مدار در برنامه HSpice از شماره ۰ تا 1E16-1 شماره‌گذاری می‌شوند. گره ۰ و GND و GND! همگی نمایانگر گره زمین هستند. ترتیب شماره‌گذاری بقیه گرهها مهم نیست. در شماره‌گذاری گرهها می‌توان بجای اعداد از اسمهای مختلفی (جهت خواناتر شدن یا معنی دار شدن اسم گرهها) از جمله in1, in2, out1, out3, insx, outsx وغیره استفاده کرد.

تمام گره‌ها باید از طریق یک مسیر dc قابل اتصال به زمین باشند. این مسیر می‌تواند از طریق مقامات‌ها، منابع تغذیه، سلف‌ها، دیودها و ترانزیستورها ایجاد شود. اگر در یک مدار گره‌ای وجود داشته باشد که هیچ مسیر dc به زمین نداشته باشد، می‌توان با اضافه نمودن یک یا چند مقاومت بزرگ به نقطه مناسب مدار آن مسیر را ایجاد کرد. بدیهی است که چون مقاومت اضافی بسیار بزرگ انتخاب می‌شود دخالتی در طرز کار مدار نخواهد داشت. برای مثال در شکل ۱، گره ۴ و یا ۳ هیچ مسیر dc به زمین ندارد. با اضافه نمودن یک مقاومت بزرگ موازی با حازن C<sub>2</sub> این مسیر ایجاد می‌گردد.



شکل ۱

با استفاده از دستور GLOBAL. می‌توان گره مورد نظر را در کل مدار مشترک کرد. اگر گرهی بصورت GLOBAL تعریف شود آن گره در تمامی زیر برنامه‌ها و برنامه، یک گره در نظر گرفته می‌شود. مثال:

.Global Vdd Vcc Clock

در تعریف گره‌های داخل زیر مدار یا ماکرو به عنوان یک خروجی با نقطه متمایز می‌شوند. و به فرم کلی زیر پیان می‌شود.

نام گره. نام زیرمدار : مثال: گره یک از زیرمدار X1.1

اگر در یک مدار بزرگ چند تا زیر مدار داشته باشیم و بخواهیم مقادیر ولتاژ یا جریان گره‌های داخل زیر مدار I را آدرس دهی کنیم باید به صورت زیر بنویسیم :

V(XAND .X1.1)	XAND	مثالاً : ولتاژ گره ۱ از زیر مدار X1 در زیر مدار
---------------	------	-------------------------------------------------

تخصیص مقدار به المانها یا پارامترهای آنها

مقادیر اجزاء مدار در برنامه HSpice می‌توانند اعداد صحیح مثبت و منفی و یا اعشاری و یا نمایی باشند. برای مثال عددی مثل  $12\text{E}-4$  ۳/۱۴ قابل قبول‌اند. می‌توان از حروف لاتین بعنوان پسوند با اعداد برای بیان ضریب عدد استفاده کرد. جدول زیر حروف مربوطه و معادل آنها را نشان می‌دهد.

F	$10^{-15}$	femto	K	$10^3$	kilo
P	$10^{-12}$	pico	MEG	$10^6$	Mega
N	$10^{-9}$	nano	G	$10^9$	Giga
U	$10^{-6}$	micro	T	$10^{12}$	tera
M	$10^{-3}$	mili			

**تیصره:** حرف m یا M برای mili است و برای  $10^{-6}$  باید از MEG استفاده شود.

در HSpice حروف زیر برای نشان دادن واحدهای مشخص بکار می‌روند.

A: ampere      Hz: hertz      H: henry      DEG: degree  
V: Volts      OHM: ohm      F: farad

روش بیان المانها

در برنامه HSpice هر نوع عنصر یا المان مدار دارای یک اسم است که این اسم با یک حرف مشخص شروع می‌شود. جدول زیر اولین حرف در نامگذاری اجزاء مدار را برای المان‌های مختلف مشخص می‌کند.

Symbols of Circuit Elements and Sources	
First Letter	Circuit Elements and Source
B	Buffer
C	Capacitor
D	Diode
E	Voltage Dependent Voltage source
F	Current- Dependent Current Source

G	Voltage Dependent Current Source
H	Current Dependent Voltage Source
I	Independent Current Source
J	JFET and MESFET Transistor
K	Mutual Inductor
L	Inductor
M	MOS Field Effect Transistor
Q	Bipolar Junction Transistor
R	Resistor
T, U, W	Transmission Line
V	Independent Voltage Source
X	Subcircuit call

اسم هر عنصر می تواند دارای ۱۶ کاراکتر مختلف باشد و از آن ۱۶ کاراکتر حرف اول حتماً باید مطابق جدول قبل انتخاب گردد بقیه کاراکترها می توانند شامل حرف و عدد باشند که بطور دلخواه انتخاب می گردند. در اسم گذاری تفاوتی بین حروف کوچک و بزرگ وجود ندارد.

### فرمت نوشتن المانهای دو سر

بطورکلی فرمت نوشتن یک عنصر دو سر از یک مدار در برنامه HSpice به صورت شکل زیر است.

مقدار N+ N- نام المان

N+ و N- شماره دو گرهای هستند که المان بین آن دو گره قرار گرفته است و فرض بر اینست که جریان از گره N+ به سمت گره N- جاری است.

### بیان مدل برای المان

گاهی مقدار یک المان، تابع پارامترهای مختلفی است در این گونه موارد پس از تعیین محل المان در مدار، اثر پارامترها را با یک "اسم مدل" و سپس نوع مدل را توسط دستور Model تعیین می کنند. در اینگونه موارد فرمت کلی بیان المان به شکل زیر است.

نام مدل [مقدار] نام المان N+ N-

برای بیان دقیق یک المان در HSpice از دستور MODEL استفاده می شود. می توان از "یک نام مدل" و یک دستور MODEL برای چندین عناصر مشابه در مدار استفاده کرد. شکل کلی دستور MODEL به صورت زیر است .

.MODEL MNAME TYPE P<sub>1</sub>=X<sub>1</sub> P<sub>2</sub>=X<sub>2</sub> ... P<sub>n</sub>=X<sub>n</sub>

MNAME اسمی انتخابی است که بعنوان "نام مدل" برای المان انتخاب می‌شود . TYPE مشخص کننده نوع المانی است که میخواهیم آنرا مدل کنیم. جدول زیر نحوه انتخاب TYPE را بیان می‌کند.  $P_1$  ,  $P_2$  ...  $P_n$  پارامترهای مربوط به المان و  $X_1$ ,  $X_2$  ...  $X_n$  به ترتیب مقادیر آن پارامترها می‌باشند.

Type name of elements	
Type name	Elements
AMP	operational amplifier model
C	capacitor model
CORE	magnetic core model
D	diode model
L	magnetic core mutual inductor model
NJF	n-channel JFET model
NMOS	n-channel MOSFET model
NPN	npn BJT model
OPT	optimization model
PJF	p-channel JFET model
PLOT	plot model for the .GRAPH statement
PMOS	p-channel MOSFET model
PNP	pnp BJT model
R	resistor model
U	lossy transmission line model (lumped)
W	lossy transmission line model
SP	S parameter

: مثال

D1 2 6 D914

.MODEL D914 D IS=1E-15 VJ=0.6 CJA=1.2E-13 CJP=1.3E-14  
+ RS=1E+8 BV=70V

در اینجا دیود D1 بین گره‌های ۲ و ۶ قرار گرفته است و دارای "نام مدل" بنام D914 است. پس از اینکه برنامه HSpice این سطر را خواند دنبال سطربی خواهد گشت که در آن مقادیر پارامترهای دیود مشخص شده باشد و آن سطربی است که با دستور MODEL. شروع می‌شود و در آن دستور دیود با نام مدل D914 تشریح شده است. حرف D برای تعیین المان مدل شده (برای دیود) بکار رفته است .

## منابع ولتاژ و جریان در HSpice

منابع ولتاژ و جریان در HSpice به دو دسته مستقل و وابسته تقسیم می‌شوند. منابع مستقل شامل منابع dc، منابع ac، منابع پالسی Pulse، منبع پاره خطی (PWL)، منبع FM (مدوله شده با یک tone)، منبع چند جمله‌ای و منبع نمائی و منبع Mixed می‌باشند.

### منابع مستقل

#### منبع ولتاژ یا جریان dc

منبع ولتاژ و جریان DC در HSpice به ترتیب با صورت کلی زیر بیان می‌شوند.

$Vx\ N+\ N-\ Dc$  مقدار یا  $Vx\ N+\ N-\$  مقدار

$Ix\ N+\ N-\ Dc$  مقدار یا  $Ix\ N+\ N-\$  مقدار

$N+$  و  $N-$  به ترتیب سر مثبت و منفی منبع می‌باشند. نمایش منابع جریان نیز مانند منابع ولتاژ می‌باشد با این تفاوت که واحدها بر حسب آمپر و اسم منابع با I شروع می‌شود.

مثال برای منبع ولتاژ:  $VCC\ 15\ 0\ DC\ 10V$  یعنی منبع ولتاژ VCC بین گره 15 و زمین قرار گرفته و مقدار آن 10V است. کلمه DC قابل حذف است ضمناً می‌توان بصورت  $DC=10$  نیز نوشت.

مثال برای منبع جریان:  $A\ 15\ 0\ DC\ 10mA$  یعنی منبع جریان Ims بین گره 15 و زمین قرار گرفته و مقدار آن  $10mA$  است.

#### منبع ولتاژ یا جریان AC

این نوع منابع در تجزیه و تحلیل Ac و نویز بکار می‌روند منبع ولتاژ و جریان AC در HSpice به ترتیب با صورت کلی زیر بیان می‌شود.

$Vx\ N+\ N-\ AC$  فاز مقدار

$Ix\ N+\ N-\ ac$  فاز مقدار

مثال برای بیان  $Vx = A \sin(wt + q)$  در بین گره‌های 1 و زمین بصورت زیر می‌نویسیم:

$$Vx\ 1\ 0\ ac = A, q \quad or \quad Vx\ 1\ 0\ ac\ A\ q$$

#### منبع ولتاژ یا جریان AC با آفست DC

اگر منبعی دارای مقدار AC و DC داشته باشد به صورت زیر بیان می‌شود.

$Vx\ N+\ N-\ DC\ AC$  فاز مقدار مقدار

$Ix\ N+\ N-\ DC\ AC$  فاز مقدار مقدار

مثال برای بیان  $Vx = A + B \sin(wt + q)$  در بین گرههای ۱ و زمین بصورت زیر می‌نویسیم :

$Vx\ 1\ 0\ DC = A\ ac = B, q$  or  $Vx\ 1\ 0\ DC\ A\ ac\ B\ q$

### منبع ولتاژ یا جریان سینوسی

این نوع منابع برای تجزیه و تحلیل گذرا (transient) مدارها استفاده می‌شوند. و به ترتیب دارای شکل کلی زیر می‌باشند.

$Vx\ N+\ N-\ SIN\ V_0\ VA\ FREQ\ TD\ ALPHA\ TETA$

$Ix\ N+\ N-\ SIN\ I_0\ IA\ FREQ\ TD\ ALPHA\ TETA$

و  $N-$  شماره دو گرهای هستند که منبع  $Vx$  یا  $Ix$  بین آن دو گره قرار گرفته است. عبارتهای بالا به ترتیب از روی روابط زیر نوشته شده‌اند.

$$V(t) = V_0 + V_A e^{-\alpha(t-td)} \sin[2pf(t-td) + q]$$

$$I(t) = I_0 + I_A e^{-\alpha(t-td)} \sin[2pf(t-td) + q]$$

که در معادله ولتاژ  $V_0$  سطح ولتاژ DC منبع سینوسی،  $V_A$  دامنه منبع سینوسی،  $f$  = FREQ تأخیر زمانی،  $\alpha$  = ALPHA ضریب تضعیف دامنه و  $\theta$  = TETA فاز آن می‌باشند.  
مثال:

$Vin\ 1\ 0\ Sin\ 0\ 1v\ 1kHz$

که معرف یک منبع ولتاژ سینوسی بین گره ۱ و زمین با دامنه ۱ ولت و فرکانس ۱ کیلو هرتز (kHz) می‌باشد.

تبصره : اگر مقادیر TD و ALPHA و TETA در رابطه تعیین نگردند، Default آنها صفر خواهد بود.

مثال دیگر: فرض کنید بخواهیم منبع  $V(t) = 1 + 10 \sin(2p \times 10^4 t + 30^\circ)$  (بین گره ۳ و ۴) را در برنامه بنویسیم این منبع بصورت زیر نوشته می‌شود.

$Vin\ 3\ 4\ Sin\ 1V\ 10V\ 1e4Hz\ 0\ 0\ 30$

و یا اگر بخواهیم منبع  $V(t) = 2 - 3e^{-10(t-20 \times 10^{-6})} \sin(2p \times 10^4 (t - 20 \times 10^{-6}))$  (بین گره ۱۰ و زمین) را بنویسیم این منبع بصورت زیر نوشته می‌شود.

$Vin\ 10\ 0\ Sin\ 2\ 3\ 10kHz\ 20US\ 10\ 180$

### منبع ولتاژ یا جریان پالسی

این نوع منابع برای تجزیه و تحلیل گذرا (transient) مدارها استفاده می‌شوند. و به ترتیب دارای شکل کلی زیر می‌باشند .

$Vx\ N+\ N-\ Pulse\ V_1\ V_2\ TD\ TR\ TF\ PW\ PER$

$Ix\ N+\ N-\ Pulse\ I_1\ I_2\ TD\ TR\ TF\ PW\ PER$

مثال

Vin 1 0 Pulse 0 5V 5uS 5uS 10uS 500uS 1ms

يعنى منبع ولتاژ پالسی بين گرههای ۱ و صفر که دارای ولتاژ اولیه صفر بوده، پس از تأخیر  $TD = 5\text{uSec}$  به

Fall time ، Rise time = 5usec در ۵ ولت باقی میماند. دارای  $TW = 500\text{uSec}$  و پریود  $1\text{mSec} = 10\text{us}$

### منبع ولتاژ یا جریان پاره خطی (Piece Wise Linear)

این منابع بصورت شکل کلی زیر میباشد:

Vx N+ N- PWL T<sub>1</sub> V<sub>1</sub> T<sub>2</sub> V<sub>2</sub> T<sub>3</sub> V<sub>3</sub> ... T<sub>n</sub> V<sub>n</sub> R T<sub>d</sub>

Ix N+ N- PWL T<sub>1</sub> I<sub>1</sub> T<sub>2</sub> I<sub>2</sub> T<sub>3</sub> I<sub>3</sub> ... T<sub>n</sub> I<sub>n</sub> R T<sub>d</sub>


منبع ولتاژ در لحظه  $T_i$  دارای ولتاژ  $V_i$  میباشد و در بین دو زمان متواتی مقدار آن از خط اتصال بین دو

نقطه  $(T_i, V_i)$  و  $(T_{i+1}, V_{i+1})$  تبعیت میکند . R برای نشان دادن تعداد دفعات تکرار موج بین زمانهای  $T_d$  تا

(برای بعد از زمان  $T_n$ ) میباشد نکته اینکه  $T_d$  حداقل باید از  $T_n$  کوچکتر باشد. برای مثال منبع شکل ۲

را میتوان به صورت زیر تعریف نمود :

Vin 1 0 PWL 0 0 8us 24v 16us 32v 28us 32v 34us 12v 44us 0v



شکل ۲


### منبع ولتاژ یا جریان نمائی

این نوع منابع به ترتیب دارای شکل کلی زیر میباشد.

Vx N+ N- EXP V<sub>1</sub> V<sub>2</sub> TD1 RTC TD2 FTC  
Ix N+ N- EXP I<sub>1</sub> I<sub>2</sub> TD1 RTC TD2 FTC

که در روابط فوق TD1 زمان تاخیر اولیه، TD2 زمان تاخیر ، RTC = Rise time constant و LT1 ولتاژ شروع و LT2 ولتاژ انتهائی می باشند.

VSS N+ N- EXP 0 10V 5us 20us 100us 30us مثال شکل ۳:



شکل ۳

### منبع ولتاژ یا جریان با مدولاسیون FM

این منبع از روی رابطه زیر تعریف می گردد و بیانگر یک موج سینوسی مدوله شده با یک فرکانس می باشد.

$$V(t) = V_0 + V_A \ Sin[2pf_C t + M \ Sin2pf_S t]$$

شکل کلی این نوع منابع به صورت زیر است.

Vx N+ N- SFFM V<sub>0</sub> V<sub>A</sub> F<sub>C</sub> Mdi F<sub>S</sub>

همان M یا ضریب مدولاسیون است. Mdi

مثال :

VSX 2 0 SFFM 0 1V 30MEG Hz 15 5kHz

ضریب مدولاسیون ۱۵ درصد است. منبع جریان با مدولاسیون FM نیز بصورت فوق تعریف می شود با این تفاوت که در منبع جریان بجای V از I استفاده می شود.

ISX 2 0 SFFM 0 1mA 30MEGHz 15 5kHz

### منبع ولتاژ یا جریان با مدولاسیون AM

این منبع بر مبنای رابطه زیر که بیانگر یک موج سینوسی است که با یک موج تک فرکانس به طور فاز مدوله شده است تعریف می گردد.

$$V(t) = S_a \{O_C + Sin[2pf_m(t - T_d)] Sin2pf_c(t - T_d)\}$$

شکل کلی این نوع منابع به صورت زیر است.

Vx N+ N- AM Sa Oc Fm Fc Td

مثال:

VSX 2 0 AM 10 1V 5kHz 30MEGHZ 1m

برای منبع جریان نیز بصورت فوق تعریف می شود با این تفاوت که در منبع جریان بجای V از I استفاده می شود.

ISX 2 0 AM 10 1ma 5kHz 30MEG Hz 1m

### منبع ولتاژ چندجمله‌ای (Poly)

منبع ولتاژی است که به بصورت یکتابع چند جمله‌ای از ولتاژ یک یا چند المان بوده و دارای سه حالت زیر می باشد.

Poly(1) One-dimensional equation

Poly(2) Two-dimensional equation

Poly(3) Three-dimensional equation

شکل کلی این منابع بصورت زیر می باشد .

$$Y = P_0 + P_1 A + P_2 A^2 + \dots + P_n A^n$$

فرم نوشتمن رابطه فوق در برنامه HSpice به صورت زیر است.

Vx N+ N- Poly NC+ NC- P0 P1 P2 ... Pn

که در این رابطه NC+ و NC- گرهایی هستند که منبع ولتاژ A (و یا شاخه حامل جریان) بین آن دو گره قرار گرفته است. مثلا:

Vx N+ N- Poly(1) 10 0 0 2

که معرف  $Y=2V(10)$  است. شکل‌های پیچیده‌تری هم برای اینگونه منابع قابل تعریف است.

مثال Poly(2)

VSX 1 2 Poly(2) 2 3 1 2 1

که معرف  $Y = 1 + 2[V(2,3)] + [V(2,3)]^2$  است.

### منبع ولتاژ یا جریان Mixed

این نوع منابع برای بیش از یک نوع تجزیه و تحلیل (مثلا برای تجزیه و تحلیل Ac، نویز، تجزیه و تحلیل گذرا) بکار می‌روند و به ترتیب دارای شکل کلی زیر می‌باشند.

Vin N+ N- Vdc AC Vac Sin Vo A f.

Iin N+ N- Idc AC Iac Sin Io A f.

در رابطه بالا f فرکانس، A دامنه و  $V_o$  مقدار آفست موج سینوسی (برای تجزیه و تحلیل گذرا)، Vac دامنه و  $V_{dc}$  مدار DC موج AC (برای تجزیه و تحلیل AC و نویز) می باشند.

مثال

VIN 13 2 0.5 AC 1 SIN 0 1 1MEG

## منابع وابسته

چهار نوع منبع وابسته در HSpice تعریف می گردد که به قرار زیرند.

- ۱- منبع ولتاژ وابسته به ولتاژ  $E < \text{نام منبع} >$
- ۲- منبع ولتاژ وابسته به جریان  $H < \text{نام منبع} >$
- ۳- منبع جریان وابسته به ولتاژ  $G < \text{نام منبع} >$
- ۴- منبع جریان وابسته به جریان  $F < \text{نام منبع} >$

## منبع ولتاژ وابسته به ولتاژ

این نوع منبع برای شبیه سازی Op-Amp ایدهال، ترانس ایدهال، گیتهای منطقی، المان تاخیر، اسیلاتور کنترل شونده با ولتاژ، منبع ولتاژ پاره خطی، چند جمله‌ای و منبع ولتاژ در سطح رفتاری بکار می رود. دو فرم از این منابع بصورت زیر می باشند.

$E(\text{name}) \quad N+ \quad N- \quad NC+ \quad NC- \quad <<\text{MAX}>=\text{Val}> \quad <<\text{MIN}>=\text{Val}> \quad \text{gain}$   
 $E(\text{name}) \quad N+ \quad N- \quad \text{TRANSFORMER} \quad NC+ \quad NC- \quad k$

$N-$  و  $N+$  دو سر منبع ولتاژ وابسته و  $NC+$  و  $NC-$  سرهای ولتاژ کنترل کننده می باشند.

مثال :

Eop\_amp 2 3 14 1 MAX=+5 MIN=-5 200  
Etrans out 0 TRANSFORMER in 0 10

مثالهای قبلی به ترتیب برای Op-Amp ایدهال و TRANSFORMER ایدهال با تابع  $V(\text{out}) = V(\text{in})/10$  استفاده می شوند.

## منبع ولتاژ وابسته به جریان

این نوع منبع برای شبیه سازی گیتهای منطقی، المان تاخیر، منبع ولتاژ پاره خطی چند جمله‌ای و منبع ولتاژ در سطح رفتاری بکار می رود.

$H(\text{name}) \quad N+ \quad N- \quad \text{VCUR} \quad <<\text{MAX}>=\text{Val}> \quad <<\text{MIN}>=\text{Val}> \quad \text{gain}$

$N+$  و  $N-$  سرهای مثبت و منفی منبع ولتاژ وابسته و  $VCUR$  منبع ولتاژی است که جریان آن (I)، اندازه ولتاژ منبع ولتاژ را کنترل می‌کند.

مثال:

H1 20 10 VCUR MAX=+10 MIN=-10 1000

منبع جریان وابسته به ولتاژ

این نوع منابع برای شبیه سازی سوئیچ‌ها، خازن متغیر با ولتاژ، منبع جریان پاره خطی چند جمله‌ای و منبع جریان در سطح رفتاری بکار می‌رود. فرم عمومی این منبع بصورت زیر است.

G (name) N+ N- NC+ NC- <>MAX>=Val <>MIN=Val gain

$N+$  و  $N-$  سرهای مثبت و منفی منبع جریان است و  $NC+$  و  $NC-$  سرهای ولتاژ کنترل‌کننده می‌باشند.

مثال:

GA 12 10 3 4 500

منبع جریان وابسته به جریان

فرم متعارف این منابع بصورت زیر می‌باشد.

F(name) N+ N- VN <>MAX>=Val <>MIN=Val gain

در رابطه فوق  $N+$  و  $N-$  سرهای مثبت و منفی منبع جریان است و  $V_N$  منبع ولتاژ مفروضی است که جریان آن، اندازه منبع جریان خروجی را کنترل می‌کند.

مثال:

F1 10 5 Vin 10  
F2 13 5 VSENS MAX=+3 MIN=-3 5

نمایش کامل عناصر غیرفعال

مقاومت

مقاومتها دارای دو شکل کلی زیر می‌باشند:

R(name) N+ N- « مقدار » (الف) :

R(name) N+ N- Model Name مقدار نامی ( $R_{T0}$ ) (ب) :

R3 2 3 10k يا R1 1 2 10Ohm مثال برای حالت الف :

R3 3 4 MSRES 10k مثال برای حالت ب :

.Model MSRES R R=1 TC1=0.02 TC2=0.002

پارامترهای مربوط به یک مقاومت بر اساس رابطه زیر تعریف می‌شوند.

$$R = R_{T0} \times [1 + TC1(T - T_0) + TC2(T - T_0)^2 + \dots]$$

مقدار مقاومت در درجه حرارت اتاق ( $27^{\circ}\text{C}$ ) است. پارامتر ضریب را در دستور Model با  $R_{T0}$  نشان می‌دهند و معمولاً برابر با ۱ است.  $TC_1$  و  $TC_2$  و ... ضرائب حرارتی مقاومت (مرتبه اول و مرتبه دوم و ...) می‌باشند. جدول زیر پارامترهای مدل را برای یک مقاومت نشان می‌دهد و اگر در دستور Model. هر کدام از پارامترها تعیین نشوند برنامه بطور اتوماتیک مقدار default آنها را در نظر می‌گیرد.

Model Parameters for Resistors				
Name	Meaning	Unit	Default	
R	Resistance multiplier	-	1	
TC1	First-order temperature coefficient	$1/\text{C}^{\circ}$	0	
TC2	Second-order temperature coefficient	$(\text{C}^{\circ})^{-2}$	0	

## خازن

شکل کلی خازنها به یکی از صورت‌های زیر است.

C (name)      N+    N-      مقدار

C (name)      N+    N-      Model Name      مقدار      IC=V<sub>0</sub>

مثال برای حالت اول

CBY    3    4    10.7uF

مثال برای حالت دوم :

CBY    4    0    MSCAP    10UF    IC=5V

.MODEL    MSCAP    C    C=1    VC1=0.01    VC2=0.002    TC1=0.01

در مثال اخیر ، IC=5V شرایط اولیه خازن را بیان می‌کند (Initial Condition) . پارامترهای مدل برای خازنها بر اساس رابطه زیر بیان می‌شوند.

$$C = C_{T0} \times [1 + VC_1 \times V_0 + VC_2 \times V_0^2] [1 + TC_1(T - T_0) + TC_2(T - T_0)^2]$$

که در آن  $VC_1$  و  $VC_2$  ضرایب ولتاژی و  $TC_1$  و  $TC_2$  ضرایب حرارتی خازن می‌باشند و  $V_0$  پتانسیل اولیه خازن است که با  $IC=V_0$  تعریف و تعیین می‌شود. جدول بعد پارامترهای مدل را برای یک خازن نشان

می‌دهد. اگر در دستور Model. هر کدام از پارامترها تعیین نشوند برنامه بطور اتوماتیک مقدار آنها default می‌گیرد.

Model Parameters for Capacitors				
Name	Meaning	Unit	Default	
C	Capacitance multiplier	-	1	
VC1	First-order Voltage coefficient	Volts <sup>-1</sup>	0	
VC2	Second-order Voltage coefficient	Volts <sup>-2</sup>	0	
TC1	First-order temperature coefficient	1/C°	0	
TC2	Second-order temperature coefficient	(C°) <sup>-2</sup>	0	

## سلف

شکل ساده بیان سلفها بصورت زیر می‌باشد:

L (name) N+ N- مقدار

L (name) N+ N- Model Name مقدار IC=I<sub>0</sub>

I<sub>0</sub> بیانگر جریان اولیه سلف می‌باشد. مثال :

L1 2 3 SELF 0.1UH IC=-20mA

.Model SELF L L=1 IL1=0.1 IL2=0.002 TC1=0.02 TC2=0.001

پارامترهای مدل برای سلفها بر اساس رابطه زیر بیان می‌شوند.

$$L = \text{ضریب } L \times [1 + IC_1 \times I_0 + IC_2 \times I_0^2] [1 + TC_1(T - T_0) + TC_2(T - T_0)^2]$$

جدول زیر پارامترهای مدل را برای سلفها نشان می‌دهد و اگر در دستور Model. هر کدام از پارامترها تعیین

نشوند برنامه بطور اتوماتیک مقدار default آنها را در نظر می‌گیرد.

Model Parameters for Inductors				
Name	Meaning	Unit	Default	
L	Inductance multiplier	-	1	
IC1	First-order Current coefficient	Amper <sup>-1</sup>	0	
IC2	Second-order Current coefficient	Amper <sup>-2</sup>	0	
TC1	First-order temperature coefficient	1/C°	0	
TC2	Second-order temperature coefficient	(C°) <sup>-2</sup>	0	


## سلفهای کوپل‌زدای

شکل کلی این نوع سلفها به صورت زیر می‌باشد.

ضریب کوپل‌زدای  $L$   $\langle \text{نام سلف اول} \rangle L \langle \text{نام سلف دوم} \rangle$

مثال :


LA 3 4 10UH  
LB 2 1 40UH  
K1 LA LB 0.9



شکل ۴

اگر جای نقطه یکی از سلف‌ها مثلاً LB عوض شود و به صورت شکل ۵ درآید می‌توانیم بنویسیم.


LA 3 4 10UH  
LB 1 2 40UH  
K1 LA LB 0.9



شکل ۵

یک ترانس با سر وسط و با ضریب کوپل‌زدای ۰/۹۹۹ (شکل ۶) به صورت زیر مدل‌سازی می‌شود.

L1 1 2 0.8mH  
L2 3 4 0.5mH  
L3 5 6 0.5mH  
K12 L1 L2 0.999  
K13 L1 L3 0.999  
K23 L2 L3 0.999



شکل ۶

در صورت یکسان بودن ضریب کوپل‌زدای، می‌توان سه عبارت اخیر را بطور خلاصه در یک دستور نوشت:

KAL L1 L2 L3 0.999

## معرفی و مدل سازی و کاربرد عناصر نیمه هادی در برنامه HSpice

بطور کلی عناصر نیمه هادی را به شکل های کلی زیر در برنامه HSpice می نویسند.

D(name) N+ N- Model name <AREA> ۱- دیود

N- گره آند ، N+ گره کاتد و AREA سطح دیود می باشند .

مثال برای دیود :

D1 7 8 DIN4001

```
.Model DIN4001 D LEVEL=1 XP=0.0 EG=1.1 XOI=0.0 XOM=0.0
+XM=0.0 WP=0.0 WM=0.0 LP=0.0 LM=0.0 AF=1.0 JSW=0.0
+PB=0.65 PHP=0.8 M=0.2994 FC=0.95 FCS=0.4 MJSW=0.5 TT=2.446e-9
+BV=4.65 RS=19 IS=1.485e-11 CJO=1.09e-9 CJP=0.0 PJ=0.0 N=1.615
+IK=0.0 IKR=1.100e-2 IBV=2.00e-2
```

Q(name) NC NB NE Model <AREA> ۲- ترانزیستور BJT

Q(name) NC NB NE NS Model name <AREA>

Model name , NC , NE و NB به ترتیب شماره گره های کلکتور، بیس، امیتر و زیربنا می باشند.

برای این نوع ترانزیستورها PNP یا NPN می باشد.

مثال :

Q(name) 1 2 3 T2n2222a

```
.Model T2n2222a NPN LEVEL=1 ISS=0. XTF=1 NS=1.00000
+CJS=0 VJS=0.5 PTF=0 MJS=0 EG=1.10000 AF=1 ITF=0.5
+VTF=1 BR=40 IS=1.6339e-14 VAF=103.40529 VAR=17.77498
+IKF=1 NE=1.31919 IKR=1 ISC=3.6856e-13 NC=1.10024 IRB=4.3646e-05
+NF=1.00531 NR=1.00688 RBM=1.0000e-02 RB=71.82988 RC=0.42753
+RE=3.0503e-03 MJE=0.32339 MJC=0.34700 VJE=0.67373 VJC=0.47372
+TF=9.693e-10 TR=380.00e-9 CJE=2.6734e-11 CJC=1.4040e-11 FC=0.950
+XCJC=0.94518
```

۳- ترانزیستور JFET و MESFET

J(name) ND NG NS Model name

Model name , ND , NG , NS به ترتیب شماره گره های درین، گیت و سورس می باشند.

برای این نوع ترانزیستورها NFET یا PFET می باشد.

۴- ترانزیستور MOSFET

M(name) ND NG NS NB Model name L W AD PD AS PS

برای این نوع ترانزیستورها NMOS یا PMOS می‌باشد. Model name بدن (Substrate)، مساحت درین، محیط درین، مساحت سورس و محیط سورس می‌باشند.

M1 3 1 2 3 MOSN L=2u W=6u AD=36p PD=24u AS=36p PS=24u  
اسم مدل MOSN می باشد که باید با دستور Model. تعریف شود.

بقيه پارامترهای المانهای مدلدار از روی مدل تعیین شده خوانده می‌شود. در هر مدلی هر پارامتری دارای معنی خاص می‌باشد ولی در اکثر مدلها این پارامترها یکسان و با یک اسم بیان شده‌اند. جهت تعیین پارامترهای تمام المانهای فوق از دستور MODEL استفاده می‌شود.

## انواع متغیرها به عنوان خروجی

در هر مداری ولتاژها و یا جریانهای مختلفی را می‌توانیم به عنوان متغیرهای خروجی تعریف کنیم. شکل کلی تعریف متغیرهای خروجی به همراه مثالی در زیر ارائه شده است:

ولتاژ یک گره نسبت به زمین: (شماره گره)  $V(7)$  مثال:

اختلاف ولتاژ دو گره نسبت به هم :  $V(N_1, N_2)$  مثال :  $V(3,4)$  یا  $V(3 \quad 4)$

اختلاف پتانسیل دو سر یک المان دو سر: (نام المان دو سر)  $V$  مثال:  $V(R1)$  یا  $V(D1)$

پتانسیل یک سر از یک عنصر سه سر نسبت به زمین: (اسم المان) نام ترمیナル عنصر سه سر  $V$

مثال : پتانسیل درین ترانزیستور  $M_3$  نسبت به زمین  $VD(M3)$

اختلاف پتانسیل دو سر از یک عنصر سه سر : (اسم المان) نام دو ترمینال از عنصر سه سر V

مشخص نمودن جریان‌ها نیز مثل ولتاژها است. جریان جاری در یک المان در یک شاخه: (نام شاخه) I

$I(D_1)$   $\neq I(R_1)$   $\neq I(VX)$

جزیره ایان ورودی به ترمینال X از یک عنصر سه سر: (نام) IX

که اولی جریان ورودی به کلکتور ترانزیستور  $Q_3$  است و دومی جریان ورودی به بیس ترانزیستور  $Q_2$  است.

در تجزیه و تحلیل AC با اضافه نمودن پسوند مناسب به نام ولتاژها و جریان‌ها می‌توان اندازه، فاز،

پخش حقیقی و یا بخش موهم یک متغیر را به دست آورد.

مثال :

VM(5)	اندازه ولتاژ گره ۵
VM(4,2)	اندازه ولتاژ بین گرههای ۴ و ۲
VP(4,2)	اندازه فاز بین گرههای ۴ و ۲
VDB(6)	اندازه ولتاژ گره ۶ بر حسب dB
VCM(Q2)	اندازه ولتاژ کلکتور ترانزیستور Q2
VDSP(M6)	اندازه فاز VDS در ترانزیستور M6
VR(2,3)	بخش حقیقی اختلاف پتانسیل بین گرههای ۲ و ۳
VI(2,3)	بخش موهومی اختلاف پتانسیل بین گرههای ۲ و ۳

برای جریان نیز می‌توان تعاریف فوق را داشت.

IM(D1)	اندازه دامنه جریان در دیود D1
IP(D1)	اندازه فاز جریان در دیود D1
IR(VIN)	بخش حقیقی جریان منع ولتاژ VIN

اگر در یک مدار بزرگ چند تا زیرمدار داشته باشیم و بخواهیم مقادیر ولتاژ یا جریان گرههای داخل زیرمدار I را آدرس دهی کنیم باید به صورت زیر عمل کنیم :

V(XAND .X1.1) XAND از زیرمدار X1

## أنواع تجزيه و تحليل مدار در برنامه HSpice

پنج نوع تجزيه و تحليل مدار در برنامه HSpice قابل اجراست:

- ۱- تجزيه و تحليل dc
- ۲- تجزيه و تحليل ac
- ۳- تجزيه و تحليل گذرا
- ۴- تجزيه و تحليل فوريه
- ۵- تجزيه و تحليل نويز

## تجزیه و تحلیل dc مدار

این تجزیه و تحلیل دارای چند نوع می‌باشد که در همه انواع تمام خازنها مدار باز و تمام سلفها اتصال کوتاه در نظر گرفته می‌شوند. انواع این تجزیه و تحلیل عبارتند از:

الف - op. این دستور (تجزیه و تحلیل) نقاط کار مدار را بدست می‌آورد. نقاط کار شامل ولتاژ گرهای، جریان شاخه‌ها و توان مصرفی المانها می‌باشد. علاوه بر اینها پارامترهای مدل سیگنال کوچک عناصر فعال نیز ارائه می‌گردد. این نتایج در Lis.finename می‌باشند. شکل کلی این دستور بصورتهای زیر می‌باشد.

```
.op  
.op [format] [time] [format] [time]
```

اگر در جلو این دستور چیزی نوشته نشود نقاط کار (ولتاژ، جریان، توان المانها و پارامترهای سیگنال کوچک عناصر فعال) را می‌دهد ولی با اضافه کردن کلماتی مثل دستور زیر می‌توان یکی از موارد نقاط کار را در زمان معین بدست آورد.

ب - SENS. این دستور (تجزیه و تحلیل) حساسیت ولتاژها و یا جریان‌های مشخصی از مدار را نسبت به تغییرات پارامترهای مدار ارائه می‌کند. تغییر پارامترها با دستور Param. انجام می‌گیرد که بعد توضیح داده می‌شود. شکل کلی این دستور بصورت زیر می‌باشد.

```
.SENS <یک یا چند ولتاژ یا جریان>
```

مثال:

```
.SENS V(9) V(4,3) V(17) I(VCC)
```

ج - TF. این دستور (تجزیه و تحلیل) مقدار تابع تبدیل سیگنال کوچک، مقاومت ورودی و مقاومت خروجی مدار را بدست می‌آورد. برای مثال اگر V(in) ولتاژ ورودی و V(out) ولتاژ خروجی باشد بهره ولتاژ  $A_u$  دارای تعریف زیر خواهد بود. این نتایج در Lis.finename می‌باشند.

$$A_u = \frac{\Delta V_o}{\Delta V_i} = \frac{V(out)}{V(in)}$$

برای بیان بهره ولتاژ فوق از دستور زیر استفاده می‌کنیم:

```
.TF Vout Vin
```

شکل کلی این دستور بصورت زیر می‌باشد.

```
.TF Y X
```

که X و Y ولتاژ یا جریان نقاط مختلف می‌باشند.

د - DC. با این دستور (تجزیه و تحلیل)، مشخصه انتقالی مدار بدست می‌آید. به این معنی که ورودی بطور dc بین دو حد مشخص تغییر می‌کند و خروجی اندازه‌گیری شده و سپس منحنی خروجی بر حسب تغییرات ورودی ارائه می‌گردد. شکل ساده دستور DC، بصورت زیر است.

.DC Input Start Value Stop Value Increment Value  
Start Value مقدار شروع، Stop Value مقدار نهائی و Increment Value مقدار گام افزایش ورودی می‌باشد.

مثال :

.DC Vin -5V 10V 0.25V

می‌توان دستور DC را به فرم زیر نوشت.

مثال :

.DC <Oct or Dec> NP Start Value Stop Value

Decade نشان دهنده طرز تغییرات ورودی و NP معرف تعداد گامها در هر Octave یا Dec می‌باشد. اگر دو ورودی در مدار وجود داشته باشند که هر دو بصورت DC در حوزه مشخصی تغییر کنند می‌توانیم برای هر دو آنها با یک دستور DC این تجزیه و تحلیل را انجام دهیم.

.DC In1 Start Value1 Stop Value1 Inc. Value1 In2 Start Value2 Stop Value2 + Inc. Value2

.DC In1 <Oct or Dec> NP1 Start Val.1 Stop Val.1 In2 <Oct or Dec> NP2 + Start Val.2 Stop Val.2

مثال :

.Dc Iref dec 5 10nA 10uA Vtest dec 2 0.5 5

تجزیه و تحلیل ac (AC).

این تجزیه و تحلیل پاسخ فرکانسی را در یک پهنه‌ای باند مشخص از فرکانس بدست می‌آورد و بر حسب درجه بندی شدن محور فرکانس به یکی از سه صورت کلی زیر نوشته می‌شود. برای این نوع تجزیه و تحلیل باید یک ورودی AC یا یک ورودی Mixed تعریف شود.

.AC LIN NP FStart FStop

.AC OCT NP FStart FStop

.AC DEC NP FStart FStop

LIN به معنی درجه بندی شدن محور فرکانس از FStart تا FStop بصورت خطی، OCT بصورت لگاریتمی در مبنای هشت و DEC بصورت لگاریتمی در مبنای دهدی می‌باشد.

(Number of points) NP : اگر تجزیه و تحلیل با LIN انجام می‌گیرد تعداد فرکانس‌هایی را نشان می‌دهد که تجزیه و تحلیل در آن نقاط انجام می‌گیرد. اگر از OCT یا DEC استفاده شود NP معرف تعداد فرکانس‌هایی است که در هر Octave و یا در هر Decade مورد بررسی قرار می‌گیرد. از تجزیه و تحلیل AC. برای پهنانی باندهای باریک و از OCT. برای پهنانی باندهای متوسط و از AC DEC. برای پهنانی باند بزرگ استفاده می‌شود.

: مثال

.AC LIN 150 100Hz 300Hz	باند باریک
.AC OCT 10 100Hz 10kHz	باند متوسط
.AC DEC 10 100Hz 10MHz	باند عریض

تبصره : می‌توان با دستورهای زیر تجزیه و تحلیل AC را به ازای مقادیر مختلف یک ورودی یا به ازای دماهای مختلف انجام داد.

.AC LIN NP FStart Fstop Sweep Input <Oct or Dec> NP1 Start Value Stop Value  
.AC OCT NP FStart FStop Sweep Input <Oct or Dec> NP1 Start Value Stop Value  
.AC DEC NP FStart Fstop Sweep Input <Oct or Dec> NP1 Start Value Stop Value

: مثال

.AC dec 10 100 100k sweep Iref dec 2 10n 10u  
.AC dec 10 100 100k sweep temp 25 50 75 100

### تجزیه و تحلیل گذرا (.TRAN)

در این تجزیه و تحلیل پاسخ گذرا یک مدار در حوزه زمان برای ورودی معینی در حوزه زمان مشخص می‌گردد. دستور TRAN. بصورت دو شکل زیر می‌باشد . برای این نوع تجزیه و تحلیل باید یک ورودی Sin ، نمائی ، پاره خطی یا Mixed تعریف شود.

.TRAN TSTEP TSTOP <UIC>  
.TRAN TSTART TSTOP TSTEP <UIC>

TSTEP زمانی است که هر بار به زمان قبلی اضافه شده و در آن لحظه مدار تجزیه و تحلیل می‌گردد.

TSTOP زمان انتهای تجزیه و تحلیل است.

TSTART زمان شروع تجزیه و تحلیل است و اگر نوشته نشود برنامه به طور اتوماتیک آن را صفر فرض می‌کند.

UIC از حروف اول کلمات Use Initial Conditions اخذ شده است. اگر خازنها و سلفها دارای شرایط اولیه باشند و بخواهیم در تجزیه و تحلیل گذرا این شرایط اولیه درنظر گرفته شوند باید از کلمه UIC در داخل دستور TRAN استفاده شود.

مثال :

.TRAN 5uS 1mS

يعنى مدار را از صفر تا msec ۱ و در فواصل usec ۵ اى تجزیه و تحلیل گذرا کن.

.TRAN 5uS 1mS 200uS

يعنى مدار را بين زمان uSec ۵ تا mSec ۱ و به فواصل uSec ۲۰۰ تجزیه و تحلیل گذرا کن.

مثال : فرض کنید يك خازن و يك سلف با شرایط اولیه در مدار وجود دارد در اين صورت دستور

TRAN. به شکل زير نوشته می شود.

```
CX   3   4   10uF   IC=-2.5V
LY   10  0   0.5uH   IC= 24.3mA
.TRAN  5uS  1mS    UIC
```

تبصره : برای سلفها و خازنها فقط در رابطه با تجزیه و تحلیل گذرا باید IC=X ( شرایط اولیه ) را مشخص نمود و در رابطه با بقیه تجزیه و تحلیل ها احتیاجی به این کار نیست .

تبصره : لازم نیست شرایط اولیه مدار از طریق تعریف در سلف و خازن نوشته شود. می توان با دستور IC. به هر کدام از گره های مدار شرایط اولیه دلخواهی را نسبت داد و از برنامه خواست با توجه به آن شرایط اولیه، مدار را تجزیه و تحلیل کند. شکل کلی دستور IC. به صورت زیر است .

.IC V(1)=V<sub>1</sub> V(2)=V<sub>2</sub> V(3,4)=V<sub>3</sub> ...

تبصره : اگر از دستور IC. استفاده شود دیگر نباید در دستور TRAN. بخش UIC نوشته شود.

تبصره : می توان با دستورهای زير تجزیه و تحلیل Transient را به ازاي مقادير مختلف يك ورودي يا به ازاي دماهای مختلف انجام داد.

```
.Tran TStep TStop <UIC> Sweep Input <Oct or Dec> NP Start Value Stop Value
.Tran TStart TStop TStep <UIC> Sweep Input <Oct or Dec> NP Start Val. Stop Val.
```

مثال :

```
.Tran 5uS 1mS UIC sweep Iref dec 2 10n 10u
.Tran 5uS 1mS UIC sweep temp 25 50 75 100
```

## تجزیه و تحلیل فوریه (.FOUR)

خروچی‌های بدست آمده در تجزیه و تحلیل گذرا جداگانه به صورت جداولی موجودند. این نمونه‌های اطلاعات را می‌توان برای بدست آوردن یک سری فوریه از سیگنال خروچی مورد استفاده قرار داد. یک سیگنال پریودیک را می‌توان توسط رابطه سری فوریه به صورت زیر نشان داد.

$$V(w) = C_0 + \sum_{n=1}^{\infty} C_n \sin(nw + f_n)$$

که در آن  $w = 2\pi f$  (فرکانس هارمونی اصلی) ،  $C_0$  مقدار متوسط موج و  $C_n$  دامنه هارمونی  $n$  ام و فاز هارمونی  $n$  ام می‌باشد. HSpice با استفاده از نتایج تحلیل گذرا ضرایب ۹ هارمونیک اول از سری فوریه را بدست می‌آورد. شکل کلی دستور Four. به صورت زیر است :

.Four FREQ X<sub>1</sub> X<sub>2</sub> ... X<sub>n</sub>

X<sub>i</sub> ها ولتاژها و جریانهای خروچی هستند که مایلیم سری فوریه آنها را بدانیم و FREQ فرکانس هارمونیک اول می‌باشد.

مثال :

.Four 100kHz V(2,3) V(3) I(R1) I(VIN)

## تجزیه و تحلیل نویز (.NOISE)

همانطوریکه می‌دانیم مقاومتها و نیمه هادی‌ها در مدار Noise تولید می‌کنند و توان این نویز تابعی از پهنهای باند مدار است. در HSpice ، تجزیه و تحلیل Noise همراه با تجزیه و تحلیل AC میسر است. مقادیر نویز در خروچی تحلیل AC می‌باشد. برای هر فرکانس تمام منابع نویز مدار محاسبه شده و اثر مجموعه آنها در خروچی به صورت کل نویز خروچی وارد می‌گردد. عبارت کلی برای تجزیه و تحلیل نویز به شکل زیر است.

.NOISE V(N+,N-) Source [M]

که در این رابطه  $V(N+,N-)$  ولتاژ خروچی بین دو گره N+ و N- است. البته خروچی می‌تواند بین یک گره و زمین باشد که در این صورت  $V(N)$  را خواهیم داشت. Source نام یک منبع ولتاژ و یا منبع جریان مستقل در ورودی مدار است که نویز به صورت ولتاژ  $\bar{I}_n^2 \bar{V}_n^2$  و جریان  $I_{in}$  نسبت به آن منبع محاسبه می‌گردد.

مثال :

.Noise V(4,5) Vin

.Noise V(6) Iin

## چاپ و رسم خروجی‌های مورد نظر در تجزیه و تحلیل‌های مختلف

دستوراتی که به کمک آنها می‌توان نتایج تجزیه و تحلیلهای پنجمگانه را مشاهده نمود و خروجی‌های مورد نظر را بدست آورد یا منحنی‌های مورد نظر را رسم نمود شامل موارد زیر می‌باشند.

.PLOT ، .PROBE ، .PRINT

برای چاپ نتایج بصورت گرافیکی و عددی بکار می‌رود این اعداد با توجه به نوع تجزیه و تحلیل بصورت زوج مرتب (مثلا در تجزیه و تحلیل گذرا ولتاژ یا جریان خواسته شده در دستور PRINT، به همراه زمان) ارائه می‌گردند. از این اعداد می‌توان بعنوان داده (مثلا در Matlab) استفاده کرد و پردازش‌های مورد نظر را انجام داد.

برای رسم منحنی تغییرات پارامترهای موردنظر بکار می‌رود. PROBE. همان نقش PLOT را دارد متنه‌ی با کیفیت و امکانات بسیار پائین و به همین دلیل در مورد آن بحثی نخواهیم داشت.

دستور PRINT

نتایج تجزیه و تحلیلهای DC، AC، گذرا و NOISE را می‌توان با کمک دستور PRINT. بصورت گرافیکی و عددی بدست آورد. شکل کلی نوشتمن این دستورات بصورتهای زیر است.

.PRINT DC [ متغیرهای خروجی مورد نظر ]

.PRINT AC [ متغیرهای خروجی مورد نظر ]

.PRINT TRAN [ متغیرهای خروجی مورد نظر ]

.PRINT NOISE [ متغیرهای خروجی مورد نظر ]

مثال :

.PRINT DC V(2) V(3,5) V(R1) VCE(Q3)

.PRINT AC V(1) V(2) I(R2) I(C1)

.PRINT TRAN V(in) V(out)

تبصره ۱ : با یک PRINT. فقط اطلاعات مربوط به ۳۲ متغیر را می‌توان محاسبه و ضبط نمود و برای محاسبه و ضبط متغیرهای بیشتر باید به تعداد دستورهای PRINT. اضافه نمود.

**تبصره ۲ :** می‌توان در دستور PRINT. بجای "متغیرهای خروجی مورد نظر"، یک یا چند پارامتر دیگر را بصورت تابعی از ولتاژ یا جریانهای نقاط دیگر تعریف کرد. این پارامتر بعد از اجرا، در داخل خروجی تحلیل مربوطه به عنوان یک متغیر با نوع Params ظاهر می‌شوند که با کلیک کردن روی آن می‌توان شکل موج توابع تعریف شده را دید. البته تعریف کردن توابع (یا رسم تابعی از چند متغیر) بعد از اجرا کردن، در خود محیط Avan waves قابل اجراست و نیاز به تعریف کردن در این قسمت نیست. می‌توان برای توابع، اسم نیز در نظر گرفت مثل gain در مثال زیر.

```
.PRINT TRAN par('(20*log10(v(out)/v(in2)))')
.PRINT AC      gain = PAR('v(3)/v(2)')      PAR('v(4)/v(2)')
```

### **دستور PROBE**

برای دیدن شکل موجهای خواسته شده در تجزیه و تحلیل‌های DC ، AC و TRAN از دستور PROBE استفاده می‌شود. اگر بخواهیم تمام متغیرها را در تمام تجزیه و تحلیل‌ها مشخص شده ببینیم این دستور دارای شکل PROBE. خواهد بود. ولی اگر بخواهیم بعضی از متغیرها را در تجزیه و تحلیل خاصی مشاهده کنیم شکل کلی این دستور بصورتهای زیر نوشته می‌شود. در اینجا نیز می‌توان توابعی را بصورت پارامتر تعریف کرد.

```
.Probe DC      [ متغیرهای خروجی مورد نظر ]
.Probe AC      [ متغیرهای خروجی مورد نظر ]
.Probe TRAN    [ متغیرهای خروجی مورد نظر ]
.Probe NOISE   [ متغیرهای خروجی مورد نظر ]
```

: مثال

```
.Probe DC      V(2)    V(3,5)    V(R1)    VCE(Q3)
.Probe AC      V(1)    V(2)      I(R2)    I(C1)
.Probe TRAN    V(in)   V(out)
.Probe TRAN    par('(20*log10(v(out)/v(in2)))')
.Probe AC      gain = PAR('v(3)/v(2)')      PAR('v(4)/v(2)')
```

## آشنایی با چند دستور مفید دیگر

SUBCKT. : اگر یک مدار مشخص بعنوان یک بخش کوچک از یک مدار بزرگ چندین بار در جاهای مختلف یک مدار بزرگ تکرار شود می‌توان با دستور SUBCKT. یک زیر برنامه برای آن بخش تکرار شونده نوشت و سپس از آن زیر برنامه در نوشتن کل برنامه برای مدار استفاده کرد ( مانند Subroutine در برنامه ). زیر برنامه SUBCKT. دارای فرمتی به صورت زیر است.

### نوشتن زیر برنامه

.SUBCKT	شماره‌های چند گره
	« یک نام دلخواه برای زیر برنامه »
	توصیف زیر برنامه : برنامه مربوط به بیان المان‌های زیر مدار .....
.ENDS	« همان نام دلخواه بالایی »

نکته ۱ : شماره گرهایی که در جلو دستور SUBCKT. و یا در داخل زیر برنامه نوشته می‌شوند گرهای محلی می‌باشند یعنی می‌توان همان شماره‌ها را در داخل برنامه اصلی یا در زیر برنامه دیگری استفاده کرد.

نکته ۲ : زیر برنامه فقط شامل عناصر مدار می‌باشد و به جز دستور MODEL. از هیچ دستور نقطه‌دار نباید در زیر برنامه استفاده شود.

نکته ۳ : زیر برنامه را می‌توان در هر کجای برنامه اصلی قرار داد یا می‌توان در فایل دیگر نوشت و در داخل برنامه اصلی آن را صدا کرد. اگر زیر برنامه در فایل دیگری نوشته شود باید در اول برنامه اصلی با استفاده از دستور include. آن را فراخوانی کرد. برای مثال اگر زیر برنامه‌ای به اسم Amplifier.sp در شاخه E:\Project نوشته شده، باید آنرا بصورت زیر در برنامه اصلی فراخوانی کرد .

.include "E:\Project\Amplifier.sp"

مثال :

```
.SUBCKT STMA 5 6 0
  R1 1 2 1k
  R2 2 0 1k
  RD 1 3 10k
  C1 5 2 10u
  C2 3 6 1u
  MQ1 3 2 0 0   MOS1 W=1u L= 1u
  VCC 1 0 5dc
  .MODEL MOS1 NMOS(VT0=1 BETA=5E-4 RD=4)
.ENDS STMA
```

## طرز استفاده از زیر برنامه

پس از نوشتن زیر برنامه ( و فراخوانی آن در صورت نوشته شدن در فایل دیگر ) می توانیم با دستوری به شکل زیر از آن استفاده کنیم.

X(name) < نام انتخاب شده برای زیر برنامه > < شماره چند گره متناظر با گره های زیر برنامه >

مثال :

X1	6	7	0	STMA
X2	4	5	0	STMA

## دستور Data

با این دستور می توان برای یک پارامتر چند تا داده تخصیص داد و بعد به ازای این داده ها مدار را تجزیه و تحلیل کرد. یا به عبارتی این دستور برای ساختن Table بکار می رود. فرم این دستور بصورت زیر است.

.DATA dataname pnam1 pnam2 pnam3 ...

مقادیر برای pnam1 و pnam2 و pnam3 (با ترتیب)

.ENDDATA

طرز استفاده :

مثال :

.AC dec 10 100 10meg SWEEP DATA = dataname

.TRAN 1n 10n SWEEP DATA = dataname

مثال کلی:

.TRAN 1n 100n SWEEP DATA = devinf

.AC DEC 10 1hz 10khz SWEEP DATA = devinf

.DC TEMP -55 125 10 SWEEP DATA = devinf

.DATA devinf width length thresh cap  
+ 50u 30u 1.2v 1.2pf  
+ 25u 15u 1.0v 0.8pf  
+ 5u 2u 0.7v 0.6pf

.ENDDATA

در مثال فوق تحلیل AC و Tran (به ازای هر دما) سه بار (تعداد دادهها) انجام می گیرند.

## Param دستور

با این دستور می‌توان به یک پارامتر یک عدد را تخصیص داد و بعد، از آن پارامتر به عنوان یک ثابت در جاهای مختلف برنامه استفاده کرد. علاوه بر آن، می‌توان در دستورات چاپ و رسم نمودار عبارتی را به یک اسم نسبت داد. در این صورت، این اسم به عنوان یک متغیر در قسمت Curves (مربوط به Param در قسمت TYPE) ظاهر گردد.

مثال :

```
.PARAM P1 =0.5u    P2 = 10u
MQ1 3 2 0 0    MOS1 L=P1    W=P2
.MODEL MOS1 NMOS(VT0=1    BETA= 5E-4    RD= 4)
.prob ac Rout=param('1m/i(vtestac)')
```

## Lib دستور

با دستور LIB. فایل Text ای که شامل پارامترهای یک یا چند مدل می‌باشد ساخته می‌شود بعد بجای دستور Model. استفاده می‌گردد. این فایل می‌تواند در هر محیطی نوشته شود. فرم نوشتن Library بصورت زیر است:

مثال : ساختار فایل Library برای سه مدل مختلف

```
.LIB entryname1
entryname1 مدل پارامترهای
.ENDL entryname1
.LIB entryname2
entryname2 مدل پارامترهای
.ENDL entryname2
.LIB entryname3
entryname3 مدل پارامترهای
.ENDL entryname3
```

که نام مدل‌های مختلف entryname می‌باشد.

## طرز فراخوانی Library

جهت فراخوانی Library در داخل برنامه می‌توان از دستورهای زیر استفاده کرد.

```
.Include '<filepath> filename'
.LIB '<filepath> filename' entryname
```

مثال:

.Include 'E:\ STORE\ STMA. LIB'

## دستور .NET

این دستور برای محاسبه پارامترهای  $Z$ ,  $H$ ,  $S$ , امپدانس ورودی و امپدانس خروجی بکار می‌رود. این آنالیز یک نوع آنالیز سیگنال کوچک تحلیل AC است، لذا این آنالیز به همراه تحلیل AC انجام می‌گیرد. این دستور برای شبکه تک پورتی بصورت زیر می‌باشد.

.NET input <RIN = Val> or .NET input <Val>

مثال:

.NET VINAC RIN = 50

.NET IIN RI N = 50

و برای شبکه دو پورتی بصورت زیر است.

.NET output input <ROUT = Val> <RIN = Val>

مثال :

.NET V (10, 30) VINAC ROUT = 75 RIN = 50

.NET I(RX) VINAC ROUT = 75 RIN = 50

برای بدست آوردن پارامترهای فوق می‌توان از دستورات زیر استفاده کرد.

.PRINT AC Z11(R) Z12(R) Y21(I) Y22 S11 S11(DB) Z11(T)

.PRINT AC ZIN(R) ZIN(I) YOUT(M) YOUT(P) H11(M) H11(T)

.PRINT AC S22(M) S22(P) S21(R) H21(P) H12(R) S22(T)

حروف داخل پرانتز مخفف کلمات زیر می‌باشند.

R: real part

I: imaginary part

M: magnitude

P: phase

DB: decibel

T: group time delay

مثال :

## \*BAND PASS FILTER

.OPTIONS DCSTEP = 1 POST

C1 IN 2 3.166PF

L1 2 3 203NH

C2 3 0 3.76PF

C3 3 4 1.75PF

C4 4 0 9.1PF

L2 4 0 36.81NH

C5 4 5 1.07PF

C6 5 0 3.13PF

L3 5 6 233.17NH

```

C7   6    7    5.92PF
C8   7    0    4.51PF
C9   7    8    1.568PF
C10  8    0    8.866PF
L4   8    0    35.71NH
C11  8    9    2.06PF
C12  9    0    4.3PF
L5   9    10   200.97NH
C13  10   OUT  2.97PF
RX   OUT  0    1E14
VIN  IN   0    AC   1
.AC  LIN  41   200MEG   300MEG
.NET V(OUT)      VIN   ROUT = 50   RIN = 50
.PLOT AC  S11(DB)   S11(P)
.PLOT AC  ZIN(M)   ZIN(P)
.END

```

## تعیین درجه حرارت

اینکار توسط دستور TEMP. انجام می‌گیرد. و شکل کلی آن بصورت مقابل است:

.TEMP <  $C^{\circ}$  > چند مقدار برای حرارت برحسب

.TEMP 25 50 125 مثال:

25

اگر مقدار درجه حرارت‌های ذکر شده از یکی بیشتر باشد مدار به تعداد درجه حرارت‌های ذکر شده بطور مجزا تجزیه و تحلیل می‌گردد.

## فرمت نوشتن فایل "Netlist file" برای یک مدار

برای تجزیه و تحلیل یک مدار باید آنرا به صورت یک فایل قابل فهم برای برنامه HSpice بنویسیم. فایل مربوطه معمولاً توسط یک برنامه Editor مانند آنچه در خود برنامه HSpice موجود است (Edit Netlist) و یا در یک Editor خارجی نوشته و در آخر برای اجرا در HSpice فایل نوشته شده را باید با پسوند sp، مثلا Adder.sp ذخیره کرد.

ترتیب نوشتن دستورات (به غیر از Title) در اجرای برنامه فرقی ندارد ولی بهتر است برنامه (file) به صورت زیر نوشته شود تا دنبال کردن و فهم برنامه راحت‌تر شود.

1-Titel

سطر عنوان برنامه

اولین سطر فایل یک عنوان اختیاری در رابطه با مدار و معرفی آن است. معمولاً اسم مدار را می‌نویسند. محتوی این سطر هیچگونه تغییری در نتیجه برنامه نخواهد داشت. مثلاً فرض کنید اگر در حال تجزیه و تحلیل گذرای یک جمع کننده هستیم، بهتر است Adder TRANSIENT ANALYSIS PROGRAM در تجزیه و تحلیل AC یک تقویت کننده فرکانس بالا، RF Amplifier AC ANALYSIS PROGRAM را در سطر اول برنامه بنویسیم. بعد از باز کردن برنامه در Hspice این عبارت در جلو Title نوشته می‌شود همچنین بعد از اجرا کردن برنامه و کلیک کردن روی Avanwaves این Title در جلو تجزیه و تحلیلها نوشته می‌شود.

## 2- Include files

در این قسمت فایلهای Library و زیر برنامه‌هایی که در جای دیگر نوشته شده‌اند را فراخوانی می‌کنیم.  
مثال :

.include 'e:\HModel3Vp6um.txt'

این دستور معرف فراخوانی یک فایل Library است که شامل مدلها می‌باشد.

مثال :

.include "E:\p5\File1.sp"

این دستورها معرف فراخوانی یک فایل از E:\p5\ می‌باشد.

## 3-Circuit descriptions

### توضیحات درباره مدار

در این قسمت ابتدا عناصر (به همراه مدل در صورت نیاز) و بعد زیر برنامه‌ها بیان می‌شود جهت اطلاع بیشتر به مثالهای آخر جزو مراجعه شود.

مثال :

```
C1    1    2    10p
R1    3    2    1k
L1    4    3    2u
m1    5    2    0    0    nlvt    l=3.5u    w=1.6u
m2    6    3    5    0    plvt    l=3.5u    w=1.6u
.SUBCKT  STMA  5    6    0
R1    1    2    1k
R2    2    0    1k
RD    1    3    10k
C1    5    2    10u
C2    3    6    1u
M1    3    2    0    0    MOS1
```

```

VCC 1 0 5dc
.MODEL MOS1 NMOS(VT0=1 BETA=5E-4 RD=4)
.ENDS STMA
X1 6 7 0 STMA
X2 4 5 0 STMA

```

#### 4-Input and Supply descriptions

در این قسمت با توجه به نوع تجزیه و تحلیل، سیگنالهای ورودی بیان می‌شوند. بهتر است ورودیهای هر زیربرنامه‌ها در داخل خود زیربرنامه مربوطه تعریف شوند.

مثال :

```

Vcc 1 0 DC 5
Iref 1 7 DC 10u
Vtest 6 9 DC 3.5
Vtestac 9 0 ac 0.5
VCin Cin 0 pulse 3.3 0 0 1n 1n 45u 90u
Vin1 10 0 Sin 2 3 10kHz 20US -10 180
Vin2 13 2 0.5 AC 1 SIN 0 1 1MEG

```

#### 5-Data and Param descriptions

در این قسمت داده‌ها و پارامترهای مورد نیاز تعریف می‌شوند.

مثال :

```

.Param L0=10u
.param Av=20*log10(v(out)/v(in2))
.DATA devinf width length thresh cap
      + 50u   30u   1.2v   1.2pf
      + 25u   15u   1.0v   0.8pf
      + 5u    2u    0.7v   0.6pf
.ENDDATA

```

#### 6-Analysis description

توضیحات درباره انواع تجزیه و تحلیل‌های موردنظر

در این قسمت نوع تجزیه و تحلیل‌ها نوشته می‌شوند.

مثال :

```

.op
.DC Vin -5V 10V 0.25V
.TRAN 5uS 1mS UIC
.Noise V(6) Iin
.AC dec 10 100 10meg SWEEP DATA = dataname

```

```
.TRAN 1n 10n SWEEP DATA = dataname
```

### 7-Output descriptions

در این قسمت دستورهای لازم برای دیدن شکل موجها نوشته می‌شوند.

مثال :

```
.prob
.prob ac Rout=par('V(9)/i(vtestac)')
```

پایان برنامه که همواره با دستور .END همراه است.

نکته ۱ : آخرین سطر از برنامه باید حتماً با دستور .END خاتمه یابد.

نکته ۲ : ترتیب بقیه سطرهای برنامه از اهمیت برخوردار نبوده و تغییری در نتیجه نخواهد داشت.


مثال :

\*\* Project2 \*\*

```
.op
.OPTIONS LIST NODE post
.include 'e:\HModel3Vp6um.txt'
```

```
vcc 1 0 DC 5
iref 1 7 Dc 10u
vtest 6 9 Dc 3.5
vtestac 9 0 ac 0.5
```

```
m1 5 2 0 0 nlvt L=0.6u W=1.5u
m2 6 3 5 0 nlvt L=0.6u W=1.5u
m3 2 2 0 0 nlvt L=0.6u W=1.5u
m4 7 7 2 0 nlvt L=12.6u W=1.5u
m5 1 7 3 0 nlvt L=1.5u W=1.5u
m6 3 2 0 0 nlvt L=2.4u W=1.5u
```



```
.tran 0.1m 1m
```

```
.ac dec 10 100 100k sweep iref dec 2 10n 10u * for Iout frequency responde if Iref
```

\*sweep between 10nA to 10uA

```
.dc iref dec 25 10nA 10uA * for Iout=function(iref)
```

```
.dc vtest dec 25 0.5 5 * for output swing or Iout=func.(Vout)
```


```
.noise v(6) iref 2
```

```
.prob ac Rout=par('V(9)/i(vtestac) ') * for output Resistance
```


```
.prob
```

```
.end
```

بعد از اجرا کردن برنامه Hspice پنجره‌ای بصورت زیر ظاهر می‌شود که در آن Open برای باز کردن فایل، Simulate برای شبیه‌سازی، Avanwaves برای دیدن شکل موجها، Edit listing برای دیدن خروجی برنامه و دیدن خطاهایی که در موقع اجرای برنامه رخ داده می‌شود) و Edit Netlist برای نوشتן فایل باز شده می‌باشند.



بعد از اجرای Avanwaves دو پنجره جدید به نامهای Avanwaves (برای رسم شکل موجها) و Results Browser (برای انتخاب موج لازم جهت رسم) به شکلهای بعد باز می‌شوند. پنجره Results Browser دارای قسمتهای مختلفی است که در جلو Design اسم و مسیر فایل اجرا شده و در زیر آن تجزیه و تحلیلهای موجود در برنامه نوشته می‌شوند. در قسمت Hierarchy کلمه Top به مفهوم کل مدار یا کل برنامه و اسامی بلوکهای زیربرنامه‌های موجود در برنامه اجرا شده نوشته می‌شود. با انتخاب نوع تجزیه و تحلیل در قسمت Type عبارتهای مربوط به نوع تجزیه و تحلیل انتخاب شده نوشته می‌شود مثلاً در حالت انتخاب کردن کلمات Transient، Time و Voltages و Currents نوشته می‌شوند. سپس با انتخاب بخش مربوطه از قسمت Hierarchy و Type با توجه به نوع تجزیه و تحلیل انتخاب شده، عبارتهایی در قسمت Curves ظاهر می‌شوند که با دو بار کلیک کردن روی هر عبارتی شکل موج مربوطه به آن در پنجره Current X-Axis Results Browser به نام Avanwaves رسم می‌شود. یک قسمت دیگری در پنجره Curves آنرا انتخاب کرده و با فشردن Apply آن پارامتر به عنوان محور X انتخاب شود در قسمت Curves وجود دارد که نشان دهنده نوع محور افقی برای نمودارهای رسم شده می‌باشد. با توجه به تجزیه و تحلیل انتخاب شده دارای یک Default می‌باشد که بطور اتوماتیک انتخاب می‌شود. برای عرض کردن آن (جهت رسم یک موج بر حسب موج دیگر) هر پارامتری را که بخواهیم به عنوان محور X انتخاب شود در قسمت Curves آنرا انتخاب کرده و با فشردن Apply آن پارامتر به عنوان محور X انتخاب می‌شود.



برای رسم تابعی از یک یا چند متغیر ابتدا در پنجره Avanwaves روی آیکون  $f(x, y)$  کلیک می‌کنیم. با این کار پنجره دیگری به نام Expression Builder باز می‌شود که این پنجره دارای توابع مختلف می‌باشد. با کلیک کردن روی هر تابعی، آن تابع در قسمت Expression نوشته می‌شود. برای نوشتمن آرگومان آن، در قسمت Curves پنجره Results Browser متغیر را انتخاب کرده و با نگه داشتن کلیکهای سمت چپ و راست (Drag & Drop) آن را به جای آرگومان تابع منتقل می‌کنیم. سپس در قسمت پنجره Expression یک اسم به تابع داده و با کلیک کردن روی آیکون Apply آن اسم در قسمت Expression پایین ظاهر می‌گردد. با دو بار کلیک کردن روی آن یا Drag & Drop کردن آن به پنجره Avanwaves ، شکل موج تابع رسم می‌شود.

**www.esud83.mihanblog.com**

**email : aminnima2@gmail.com**

**amin sheikh najdi**

**tel : 0916642067**